2024-06-03
John A. Marohn (jam99@cornell.edu)

Roger Loring's new theory for the broadband local dielectric spectroscopy signal should also predict the so-called local dielectric specroscopy (LDS) signals oscillating at frequency 2 ω$m. There should be an in-phase frequency modulation, proportional to Cos[2 ω$m], and an out-of-phase frequency modulation, proportional to Sin[2 ω$m]. In the note below, retrace my 2024-05-29 paper-and-pencil derivation computing time-averaged frequency shift. Then extend the calculation to compute the LDS lock-in signals.

Following Loring's 2024-02-20 notes, define the average interaction energy. Here V$ts is the tip-sample voltage, c is the tip capacitance, and ϕ is the reaction potential.

In[174]:=
```
Clear[W, V$ts, c, ϕ];
W[z_, t_] = V$ts[t] × c[0, z] ϕ[z, t]
```

Out[175]=
```
c[0, z] × V$ts[t] ϕ[z, t]
```

Here *c*[*n*, *z*] represents the nth derivative of the tip-sample capacitance *c*[*0*, *z*]. Every time you take a derivative, you increase *n*.

In[176]:=
```
Derivative[0, 1][c][n_, z_] := c[n + 1, z]
```

As a check, take the second derivative of the interaction energy. Observe that I get the expected paper-and-pencil answer.

In[177]:=
```
D[W[z, t], {z, 2}] // Simplify
```

Out[177]= $V\$ts[t]\left(c[2, z]\,\phi[z, t] + 2\,c[1, z]\,\phi^{(1,0)}[z, t] + c[0, z]\,\phi^{(2,0)}[z, t]\right)$

Define the reaction field . Here *R*[*k, n, z, ω*] is the frequency-domain response function. The first input *k* indicates the real or imaginary part, with *k* = 0 the real part and *k*=1 the imaginary part. The second input *n* indicates the *n*th derivative. The third and fourth inputs are height and modulation frequency, respectively.

In[178]:=
```
Clear[ϕ$rxn, R, ω$m];
ϕ$rxn[z_, t_] = V$0 c[0, z] (R[0, 0, z, ω$m] Cos[ω$m t] - R[1, 0, z, ω$m] Sin[ω$m t]);
```

Every time you take a derivative of *R* with respect to *z*, increase *n* by one.

In[180]:=
```
Derivative[0, 0, 1, 0][R][k_, n_, z_, ω_] := R[k, n + 1, z, ω]
```

Check that we get the expected behavior.

In[181]:=
```
D[ϕ$rxn[z, t], z]
```

Out[181]=
```
V$0 c[1, z] (Cos[t ω$m] R[0, 0, z, ω$m] - R[1, 0, z, ω$m] Sin[t ω$m]) +
  V$0 c[0, z] (Cos[t ω$m] R[0, 1, z, ω$m] - R[1, 1, z, ω$m] Sin[t ω$m])
```

Substitute the reaction potential and the oscillating tip-sample voltage into the interaction energy.

In[182]:=
```
W$new[z_, t_] = W[z, t] /. {ϕ[z, t] → ϕ$rxn[z, t], V$ts[t] → V$0 Cos[ω$m t]}
```

Out[182]=
$$\text{V\$0}^2\, c[0,z]^2 \operatorname{Cos}[t\,\omega\$m]\,(\operatorname{Cos}[t\,\omega\$m]\, R[0,0,z,\omega\$m] - R[1,0,z,\omega\$m] \operatorname{Sin}[t\,\omega\$m])$$

In[183]:=
$$\Delta\omega = \frac{\text{w\$c}}{2\,\text{k\$c}}\,\text{D[W\$new[z, t], \{z, 2\}] // ExpandAll}$$

Out[183]=
$$\frac{\text{V\$0}^2\,\text{w\$c}\, c[1,z]^2 \operatorname{Cos}[t\,\omega\$m]^2 R[0,0,z,\omega\$m]}{\text{k\$c}} + \frac{\text{V\$0}^2\,\text{w\$c}\, c[0,z] \times c[2,z] \operatorname{Cos}[t\,\omega\$m]^2 R[0,0,z,\omega\$m]}{\text{k\$c}} + \frac{2\,\text{V\$0}^2\,\text{w\$c}\, c[0,z] \times c[1,z] \operatorname{Cos}[t\,\omega\$m]^2 R[0,1,z,\omega\$m]}{\text{k\$c}} + \frac{\text{V\$0}^2\,\text{w\$c}\, c[0,z]^2 \operatorname{Cos}[t\,\omega\$m]^2 R[0,2,z,\omega\$m]}{2\,\text{k\$c}} - \frac{\text{V\$0}^2\,\text{w\$c}\, c[1,z]^2 \operatorname{Cos}[t\,\omega\$m]\, R[1,0,z,\omega\$m] \operatorname{Sin}[t\,\omega\$m]}{\text{k\$c}} - \frac{\text{V\$0}^2\,\text{w\$c}\, c[0,z] \times c[2,z] \operatorname{Cos}[t\,\omega\$m]\, R[1,0,z,\omega\$m] \operatorname{Sin}[t\,\omega\$m]}{\text{k\$c}} - \frac{2\,\text{V\$0}^2\,\text{w\$c}\, c[0,z] \times c[1,z] \operatorname{Cos}[t\,\omega\$m]\, R[1,1,z,\omega\$m] \operatorname{Sin}[t\,\omega\$m]}{\text{k\$c}} - \frac{\text{V\$0}^2\,\text{w\$c}\, c[0,z]^2 \operatorname{Cos}[t\,\omega\$m]\, R[1,2,z,\omega\$m] \operatorname{Sin}[t\,\omega\$m]}{2\,\text{k\$c}}$$

To get the dc frequency shift, average over one oscillation cycle. Define a function to average over an oscillation cycle.

In[184]:=
$$\text{cycleAverage[f\_]} := \frac{\omega\$m}{2\,\pi}\,\text{Integrate}\left[f, \left\{t, 0, \frac{2\,\pi}{\omega\$m}\right\}\right]$$

Check that averaging 1 over an oscillating cycle returns 1 and check that averaging cosine squared gives 1/2. These checks show that I have the prefactor correct in the above function definition.

In[185]:=
```
{cycleAverage[1], cycleAverage[Cos[ω$m t]^2]}
```

Out[185]=
$$\left\{1, \frac{1}{2}\right\}$$

Define another function to factor out a common factor, from here.

In[186]:=
```
factorOut[fac_][expr_] := Replace[expr, p_Plus :> fac Simplify[p / fac], All]
```

Now cycle-average the frequency shift, collect the prefactors of the three R derivatives, and factor out a common term.

In[187]:=
```
Δω$avg = factorOut[V$0^2 w$c / (4 k$c)][Collect[cycleAverage[Δω],
    {R[0, 0, z, ω$m], R[0, 1, z, ω$m], R[0, 2, z, ω$m]}, FullSimplify]]
```

Out[187]=
$$\frac{1}{4\,k\$c} V\$0^2\, w\$c\, \big(2\, c[1, z]^2\, R[0, 0, z, \omega\$m] + 4\, c[0, z] \times c[1, z] \times R[0, 1, z, \omega\$m] + c[0, z]\, (2\, c[2, z] \times R[0, 0, z, \omega\$m] + c[0, z] \times R[0, 2, z, \omega\$m])\big)$$

This result agrees with my and Loring's paper-and-pencil finding.

To obtain the LDS signals, mimic lock-in detection at a reference frequency of 2 *ω*$m. Multiply the frequency shift by Cos[2 *ω*$m *t*] and Sin[2 *ω*$m *t*] before performing the cycle average.

In[188]:=
```
LDS$X = factorOut[V$0^2 w$c / (8 k$c)][Collect[cycleAverage[Δω Cos[2 ω$m t]],
    {R[0, 0, z, ω$m], R[0, 1, z, ω$m], R[0, 2, z, ω$m]}, FullSimplify]]
```

Out[188]=
$$\frac{1}{8\,k\$c} V\$0^2\, w\$c\, \big(2\, c[1, z]^2\, R[0, 0, z, \omega\$m] + 4\, c[0, z] \times c[1, z] \times R[0, 1, z, \omega\$m] + c[0, z]\, (2\, c[2, z] \times R[0, 0, z, \omega\$m] + c[0, z] \times R[0, 2, z, \omega\$m])\big)$$

In[189]:=
```
LDS$Y = factorOut[-V$0^2 w$c / (8 k$c)][Collect[cycleAverage[Δω Sin[2 ω$m t]],
    {R[1, 0, z, ω$m], R[1, 1, z, ω$m], R[1, 2, z, ω$m]}, FullSimplify]]
```

Out[189]=
$$-\frac{1}{8\,k\$c} V\$0^2\, w\$c\, \big(2\, c[1, z]^2\, R[1, 0, z, \omega\$m] + 4\, c[0, z] \times c[1, z] \times R[1, 1, z, \omega\$m] + c[0, z]\, (2\, c[2, z] \times R[1, 0, z, \omega\$m] + c[0, z] \times R[1, 2, z, \omega\$m])\big)$$

Observe that the in-phase lock-in signal at 2 *ω*$m is just *half* the average frequency shift. This is a somewhat surprising result. I expected the in-phase lock-in signal to involve different combinations of capacitance and response-function derivatives.

In[190]:=
```
LDS$X / Δω$avg // FullSimplify
```

Out[190]=
$$\frac{1}{2}$$

The out-of-phase lock-in signal is distinct from the in-phase signal, since the out-of-phase signal involves derivatives of the *imaginary* part of the response function R.